
Chapter 1

Disjoint Forms in Graphical
User Interfaces
Sander Evers, Peter Achten, Rinus Plasmeijer1

Abstract: Forms are parts of a graphical user interface (GUI) that showa (struc-
tured) value and allow the user to update it. Some forms express a choice between
two or more (structured) values using radio buttons or checkboxes. We show that
explicitly modelling such a choice leads to a cleaner separation of logic and lay-
out. This is done by extending the combinator libraryFunctionalForms with dis-
joint form combinators. To implement these, we have generalized the technique
of compositional functional references which underlies the library.

1.1 INTRODUCTION

Formsare parts of a graphical user interface (GUI) that show a (structured) value
and allow the user to update it. For example, the omnipresentdialogs labeled
Options, SettingsandPreferencesare forms. An address book can also be con-
sidered a form. In our previous work, we have developed the combinator library
FunctionalForms[2] for building forms in a concise, compositional way.

Many real-life forms allow a choice between two or more alternatives, some
of which require extra information. For example, the form inFig. 1.1 indicates
whether the user wishes to receive a certain newsletter; if s/he does, the text entry
field next to this option should contain his/her email adress. If s/he does not, this
text field is irrelevant (some GUIs provide a visual clue for this: the control is
dimmed).

Usually, the information in such a form is processed as a product-like data
structure containing the choice (e.g. as a boolean) and the extra information (e.g.
as a string). However, most functional languages allow datatypes which are more
suited to this task, namelydisjoint union types. In Haskell[4], we would define

FIGURE 1.1. A ‘disjoint’ form

data NewsLetter= NewsYes String|NewsNo

for the type of information in the example form.
While the combinators inFunctionalForms previously only supported forms

with product-like data structures, in this paper we extend them to enable the ex-
plicit definition of such adisjoint form. Rather than as a ‘yes/no’ formand an
‘email’ form, it can now be composed as a ‘yes, email’ formor a ‘no’ form. We
demonstrate that this technique leads to a better separation of logic and layout in
disjoint forms. For its implementation, we have generalized thecompositional
functional referenceswhich underlie the library.

This paper is organized as follows: it first gives a summary ofthe library’s
basic use, which has not changed (Sect. 1.2). Then, the use and merits of the ex-
tension are demonstrated (Sect. 1.3), after which its implementation is discussed
in Sect. 1.4. Next, we show that the gained flexibility leads to some safety issues
(Sect. 1.5). We finish with related work (Sect. 1.6) and conclusions (Sect. 1.7).

1.2 FUNCTIONALFORMS SUMMARY

FunctionalForms[2] is a combinator library built on the GUI librarywxHaskell[7]
(itself built on the cross-platformC++ library wxWidgets[11]). It can be seen as
an embedded domain-specific language for forms: it consistsof atomic formsand
ways to combine them into larger forms in a declarative style. In this section,
we give a brief summary of its basic use, which is the same as described in [2],
although the types have changed a little.

A form is a GUI part, residing somewhere within a dialog withOK andCancel
buttons, which is only able to display and alter a certain value. When the dialog
appears, the form has aninitial value which is provided by its environment; sub-
sequently, the user can read and alter this value; at the end,the user closes the
dialog with one of the buttons, and the form passes thefinal valueback to the
environment. The type of this value is called thesubject typeof the form.

Atomic forms correspond to a single control containing an editable value. Ex-
amples are a text entry field, containing aString, and a spin control, containing an
Int:

entry′ :: Monad m⇒
[Prop(TextCtrl())] → Ref m String→ FForm win m Layout

spinCtrl ′ :: Monad m⇒
Int → Int → [Prop(SpinCtrl())] → Ref m Int→ FForm win m Layout

FIGURE 1.2. ticketsForm

We follow the convention that library functions are underlined, and an atomic
form is named after the correspondingwxHaskell function which creates its con-
trol, but with an additional prime symbol. Every atomic formis parameterized
with a list of optional properties used for customizing the control, e.g. its size and
font (leaving this list empty produces reasonable defaults). Some atomic forms,
like spinCtrl ′, require additional parameters: its first two arguments indicate a
minimum and maximum value.

All these parameters actually have little to do withFunctionalForms: they are
directly passed to thewxHaskell control creation function. In contrast, the last
parameter of both forms is specific toFunctionalForms; it is a reference value
which relates the atomic form’s subject type (StringandInt, resp.) to the subject
type of the top-level form. A more detailed description of the Ref andFForm
types is postponed to Sect. 1.4.

To combine atomic forms into larger forms, two aspects have to be composed:
layout and subject type. The former is performed bylayout combinatorslike grid′,
margin′ andfloatLeft′. These are based on thewxHaskell layout combinators after
which they are named, but operate directly on forms.1 For example, the two
atomic forms can be put in a grid layout with some labels (see Fig. 1.2):

grid′ 5 5 [ [ label′ ”name :”, entry′ [] name ]
, [ label′ ”nr. of tickets :”, spinCtrl ′ 1 6 [] nr ]
]

Note that the two reference values(name :: Ref m String) and(nr :: Ref m Int)
are free variables in this expression. Also, this form’s subject type is not yet es-
tablished. To complete the form composition,nameandnr are bound in a lambda
expression, onto which asubject type combinator, namelydeclare2, is applied:

ticketsForm:: Monad m⇒ Ref m(String, Int) → FForm win m Layout
ticketsForm= declare2$ λ(name,nr) →

grid′ 5 5 [ [ label′ ”name :”, entry′ [] name ]
, [ label′ ”nr. of tickets :”, spinCtrl ′ 1 6 [] nr ]
]

This ‘declares’ticketsForm’s subject type to be(String, Int), as witnessed by
its actual type declaration (which can be omitted). Just like the atomic forms,
ticketsFormcan now be used as a component of a larger form. Note how this two-
stage process of form construction separates the definitionof layout and subject
type structures, providing a great deal of freedom to the library user (see also [2]).

1instead of onLayoutvalues of widgets—for those familiar withwxHaskell

FIGURE 1.3. contactForm1

Besidesdeclare2, which declares a pair, the library also provides subject type
combinators for tuples of higher arity and for lists.

declare2 :: Monad m⇒
((Ref m t1, Ref m t2) → z) → Ref m(t1, t2) → z

declare3 :: Monad m⇒
((Ref m t1, Ref m t2, Ref m t3) → z) → Ref m(t1, t2, t3) → z

· · ·
declareL :: Monad m⇒

([Ref m t] → z) → Ref m[t] → z

The declareL combinator only composes forms for the listelementsand cannot
alter the spine; it produces a form for lists of a fixed length.

To run a form in awxHaskell program, the library functionrunInDialog is
used. For example, this runs the above definedticketsFormwith Johnand2 in the
atomic forms:

do . . .

(newname,newnr) ← runInDialog parentWindow ticketsForm(”John”,2)
. . .

The function takes as its arguments a pointer to a parent window, the form itself,
and an initial value of the form’s subject type. It returns anIO action, which
produces a modal dialog containing the form andOK/Cancelbuttons. When the
user pressesOK, the return value is bound to altered value in the form; ifCancel
is pressed, the initial value is returned instead. After this, the IO thread continues.

1.3 COMBINATORS FOR DISJOINT FORMS

This section describes, from a library user’s point of view,the additions for defin-
ing disjoint forms. As an example, we will define a form for contact information,
depicted in Fig. 1.3. It has subject type

data Contact = ByPhone Phone| ByEmail String|NotAtAll

and expresses a choice between a phone number, an email address and no in-
formation at all. Before we can start defining the form itself, we need to define
three custom subject type combinators for this type’s data constructors. This is

done using aTemplate Haskell [8] macro nameddeclare, which is included in the
library.

declareByPhone= $(declare [|ByPhone|] 1)
declareByEmail= $(declare [|ByEmail|] 1)
declareNotAtAll= $(declare [|NotAtAll|] 0)

For each of the constructors, we provide the macro with its name and arity. The
delimiters $(. . .) and [| . . . |] areTemplate Haskell syntax, which the library user
does not need to worry about.2

The three fresh subject type combinators are used to turn forms with subject
types (resp.)Phone, Stringand no subject type at all into forms with subject type
Contact. Their type signatures are:

declareByPhone:: Monad m⇒
(Ref m Phone→ z)→ Ref m Contact→ z

declareByEmail:: Monad m⇒
(Ref m String→ z)→ Ref m Contact→ z

declareNotAtAll:: Monad m⇒
FForm win m l→ Ref m Contact→ FForm win m l

In the last type signature, the typeFForm win m lplays the same role asz in the
above signatures. The reason why it is more constrained is that declareNotAtAll
appends its argument form with an invisible form for handling theNotAtAllvalue.

Using these subject type combinators,contactForm1 can be defined as fol-
lows; we assume(phoneForm:: Ref m Phone→ FForm win m Layout) is de-
fined somewhere else:

contactForm1 = radioGrid [byPhone, byEmail, byNothing]
byPhone = declareByPhone$ λphone→

row′ 5 [label′ ”by phone :”, phoneForm phone]
byEmail = declareByEmail$ λemail→

row′ 5 [label′ ”by email :”, entry′ [] email]
byNothing = declareNotAtAll$

label′ ”do not contact me”

The new disjoint form combinatorradioGrid arranges its list of subforms into
a grid layout, with radio buttons to the left of them. Due to their subject type
combinators, the three subforms have the same subject type as the composite
form (Contact), but each only ‘works’ for a particular data constructor. For ex-
ample, thebyEmail form only handlesByEmailvalues. This means that when
contactForm1 is run with an initialByEmailvalue, the middle radio button is se-
lected, and only the text field next to it receives an initial value. The other text

2Template Haskell is a GHC compiler extension for meta-programming, i.e.
programmatically manipulating a program at the syntactic level. The delimiters turn a
meta-language expression into an object-language expression and vice versa. Both object
language and meta-language areHaskell.

FIGURE 1.4. contactForm2 FIGURE 1.5. contactForm3

field is left empty (or contains a default value, if the programmer has specified
this in phoneForm). When the form is closed, every subform contains a final
value with its particularContactdata constructor, but only one of them is pro-
moted tocontactForm1’s final value; this choice is determined by the radio button
selected at that time.

What is the advantage of using the disjoint form combinatorradioGrid, apart
from stylistic arguments? Consider the alternative case, in which the form in
Fig. 1.3 is defined as aconjunctionof a radioBox′ (with an Int for three possible
choices), aphoneFormand anentry′; its subject type would be(Int,Phone,String).
At some later time, the interaction design department decides the form should
rather look like Fig. 1.4 or like Fig. 1.5. Note that these forms still express exactly
the same choice. However, when the form code is changed accordingly, its subject
type would be(Int, Int,Phone,String) or (Bool, Int,Phone,String), and the code
which handles the form data should also be altered.

If we use disjoint forms instead, the disjoint subject type can remain the same.
In the code, we only need to add an extraradioGrid for the first case:

contactForm2 = radioGrid
[noContact, λyes→ row′ 5 [label′ ”yes”, yesContact yes]]

noContact = declareNotAtAll$ label′ ”no”
yesContact = radioGrid [byPhone, byEmail]

For the second case, we use another disjoint form combinator, namelycheckRow:

contactForm3 = checkRow
(λyes→ column′ 5 [label′ ”Please contact me” , yesContact yes])
(declareNotAtAllnoLayout)

The functionality of these forms is still the same: they display a value of their
subject typeContact, and allow the user to change it into another value of that
type.

1.4 IMPLEMENTATION

Although the user ofFunctionalForms does not notice a difference, apart from the
new combinators and slightly alteredRefandFForm types, the implementation of
the library has undergone substantial changes since its first version in [2]. These

allow for generalized forms, which mayfail to consume an initial value (or pro-
duce a final value), and which can be joined with the disjoint form combinators
radioGrid andcheckRow. To construct these forms, thecompositional functional
referenceshave also been generalized. Furthermore, the ‘heart’ of a form, which
determines the communication with its environment, has been made explicit in a
type RefLink. In order to deal with the newFForm type in Sect. 1.4.3, we will
first discuss theseRef andRefLinktypes.

1.4.1 The Ref type

A reference value consists of two functions which are used to‘refer to’ a t part of
a—usually stateful—monadm:

data Ref m t= Ref{ val :: m t
, app:: (t→m t)→m()
}

The val function retrieves the value of this particular part of the monadic state,
whereas theapp function updates it. For example, for a state of type(t1, t2), the
value referring to thet1 element would be:

reffst:: MonadState(t1, t2) m ⇒ Ref m t1
reffst= Ref{ val = do { (x,y) ← get; return x}

, app= λf → do { ˜(x,y) ← get; x′ ← f x; put(x′,y) }
}

Note the lazy pattern match in theapp function; it is useful when constructing a
new state from scratch (i.e. the previous state contains⊥).

A reference to the value in aJustdata constructor (from the well-knownMaybe
type) can be defined in a very similar way:

reffromJust:: MonadState(Maybe t) m ⇒ Ref m t
reffromJust= Ref
{ val = do { Just x← get; return x}
, app= λf → do { ˜(Just x) ← get; x′ ← f x; put(Just x′) }
}

This reference value may seem ill-defined because it can ‘dangle’: when the
monadic state containsNothing, it does not refer to anything. However, this situ-
ation can be detected using monadic error-handling, and thecontrol flow can be
adapted. We will show how this is done in Sect. 1.4.2, when we join twoRefLinks.

The operator• composes two reference values, taking the referred part of the
second value’s state as the state of the first value. For example, the following
value refers to the first element of a pair within aJustvalue:

reffst • reffromJust:: MonadState(Maybe(t1, t2)) m ⇒ Ref m t1

The composition is performed by applying a monad transformer to the monad of
the second reference value. This ‘adds state’ to this monad,on which the first

reference value can act. Meanwhile, properties of the original monad such as IO
ability or error handling are preserved.

• :: Monad m⇒ Ref (StateT cx m) t → Ref m cx→ Ref m t
w • v = Ref
{ val = val v �= evalStateT(val w)
, app= λf → (app v) $ execStateT$ (app w) $ lift . f
}

This operator is used to define subject type combinators like:

declare2 :: Monad m⇒
((Ref m t1, Ref m t2) → z) → Ref m(t1, t2) → z

declare2 refsToForm refP= refsToForm(reffst • refP, refsnd• refP)

declareJust :: Monad m⇒ (Ref m a→ z) → Ref m(Maybe a) → z
declareJust refToForm refMaybe= refToForm(reffromJust• refMaybe)

These subject type combinators all follow the same pattern.This pattern is cap-
tured in theTemplate Haskell macrodeclare, so definitions like the two above do
not have to be handwritten for every data constructor.

1.4.2 The RefLinktype

The heart of an atomic form consists of a link between two reference values.
The first is of typeRef m t, and relates the subject typet of this form to that
of the topmost form (the state type inm). This is the reference value that is
explicitly provided by the library user, e.g. in the expression entry′ [] refm. The
second reference value is implicit in every atomic form; it is of typeRef IO t, and
relates this form’s subject type to a part of the IO state. It is constructed from
wxHaskell’s getandset functions for the control’s main attribute (e.g.text on an
entrycontrol).

In the terminology of the well-known model–view(–controller) paradigm[6],
the reference values refer to a part of the topmost form’smodeland a part of its
view, respectively. Joining them in aRefLinkmeans linking those parts to each
other: theval output from the first reference is used asapp input for the second,
and vice versa. Thus, two operations are obtained which bothenforce consistency
between model and view:

• Theupdateoperation copies the value from the form’s model to its view.This
is used to show the form’s initial value.

• Thepropagateoperation copies the value from the form’s view to its model.
This is used to read the form’s final value.

In monadic terms, theupdateoperation is a read action in themmonad producing
a write action in theIO monad. Conversely, thepropagateoperation is a read

action in theIO monad producing a write action in them monad. This results in
the following type forRefLink(where then monad abstracts fromIO):

data RefLink m n= RefLink
{ update :: m (n ())
, propagate:: n (m ())
}

The functionrefLinkconnects the two references to create such aRefLink. For the
updatefunction, first an inputv is retrieved from them reference. Then, a constant
functionconst(return v) is applied to the corresponding part in then monad using
the n reference; this action in then monad is returned in them monad. For the
propagatefunction, the roles ofm andn are reversed:

refLink :: (Monad m, Monad n) ⇒ Ref m a→ Ref n a→ RefLink m n
refLink refm refn = RefLink
{ update = (val refm) �= return . (app refn) . const. return
, propagate= (val refn) �= return . (app refm) . const. return
}

When two forms are joined, theirRefLinks are combined into a newRefLink.
Usually, the intention is that the jointupdateperformsbothcomponentupdates,
and likewise for thepropagates. We consider this to be the ‘default’ operator on
RefLink. In order to meet theMonadWriterinterface (see Sect. 1.4.3), we encode
it using theMonoidclass:

instance(Monad m, Monad n) ⇒ Monoid(RefLink m n)
where mempty= RefLink

{ update = return$ return()
, propagate= return$ return()
}

rl1 ‘mappend‘ rl2 = RefLink
{ update = liftM 2 (�) (update rl1) (update rl2)
, propagate= liftM 2 (�) (propagate rl1) (propagate rl2)
}

For disjoint forms, the twoRefLinks should be joined in an alternative way. In
this situation, they share one part of the model part, which is a disjoint union (e.g.
the subject type of both forms isEither a b). Meanwhile, they refer to different
parts of the view (which contains controls for botha and b). Hence, the two
RefLinks connect independent parts of the view state space to ‘competing’ parts of
the model state space. Instead of performing bothupdate(propagate) operations,
only one can (should) be performed.

We obtain this behaviour by using themplusoperator of the model monadm;
therefore, this should be an instance ofMonadPlus. The jointupdatewill then
(dynamically) choose between the first componentupdateor the second—and
likewise forpropagate. Hence, we define an alternative monoid on theRefLink

domain. By using a different representation for theRefLinktype, theMonoidclass
can again be used for this:

newtype RefLinkPlusM m n= RefLinkPlusM{pm :: RefLink m n}
instance(MonadPlus m, Monad n) ⇒ Monoid(RefLinkPlusM m n) where

mempty= RefLinkPlusM$ RefLink
{ update = mzero
, propagate= return mzero
}

rl1 ‘mappend‘ rl2 = RefLinkPlusM$ RefLink
{ update = (update$ pm rl1) ‘mplus‘ (update$ pm rl2)
, propagate= liftM 2 mplus(propagate$ pm rl1) (propagate$ pm rl2)
}

The exact semantics ofmplusdepend on the monadm. In practice, we use an
error-handling state monadErrorT e(State a). This means that the first argument
of mplusis always tried first; if it fails, the second argument is tried. When disjoint
forms are used correctly, the alternatives are mutually exclusive, so this order is
irrelevant.

1.4.3 The FForm type

A form is a value of the following type:

newtype FForm win m a= FForm
{ runFForm :: Window win→ IO (a, RefLink m IO) }

It contains three pieces of information:

1. An IO action which creates the form’s controls. This action depends on a
pointer to a parent window of typeWindow win, in which they are created.

2. A RefLinkused to update the values in the controls from—and propagatethem
to—the form’s model.

3. Additional information of typea; usually layout information of typeLayout
(defined by thewxHaskell library).

The FForm type can be used as a monad, whichbinds the additional (layout)
information,readsthe window pointer,executesthe control creation functions,
andwritesa RefLink.

instance(Monad m) ⇒ Monad(FForm win m)
instance(Monad m) ⇒ MonadReader(Window win) (FForm win m)
instance(Monad m) ⇒ MonadIO(FForm win m)
instance(Monad m) ⇒ MonadWriter(RefLink m IO) (FForm win m)

So,form1 �= f means:

• f is applied to the additional information fromform1, producing (let’s call it)
form2.

• The window pointer passed toform1 �= f is passed toform1 andform2.

• The IO actions inform1 andform2 are sequenced.

• The RefLink in form2 is joined to theRefLink in form1 using the ‘default’
mappendoperator.

Furthermore, functions likeask(extract the window pointer),liftIO (insert an IO
action at form creation time) andtell (insert aRefLink) are implemented for the
FFormmonad (being an instance ofMonadReader, MonadIOandMonadWriter).

We stated in Sect. 1.4.2 that in order to combine two forms in adisjoint way,
the RefLinkPlusMmonoid should be used, which dynamically performsoneof
the update/propagateoperations. Meanwhile,both forms should be shown: at
form creation time,both IO actions should be performed, andboth layout values
are used. Therefore, we have implemented alternative bind and unit operators for
forms: �± and return0. They are similar to�= andreturn in every respect,
except that they use theRefLinkPlusMmonoid.

return0 :: MonadPlus m⇒ a → FForm win m a
(�±) :: MonadPlus m⇒

FForm win m a→ (a → FForm win m b) → FForm win m b

These operators are at the core of the disjoint form combinators radioGrid and
checkRow, whose implementation is discussed in the next section.

1.4.4 Disjoint form combinators

A naı̈ve disjoint form combinator would be

refToForm1 ‘or‘ refToForm2 = λref →
refToForm1 ref �± λlay1 →
refToForm2 ref �± λlay2 →
return0 $ column5 [lay1, lay2]

The composite form shows both forms, while the compositeupdatefunction per-
forms only one of the componentupdatefunctions—the first one that succeeds.
The same goes for the compositepropagateoperation. However, the form’s user
has no means whatsoever to discover whichupdatehas been performed, or to
influence whichpropagateto perform!3

TheradioGrid combinator does provide these functions: both are fulfilledby
the radio buttons. When a subform’supdateis performed, the system selects the
radio button in front of it. Conversely, the form’spropagateis only performed if

3Bothpropagateoperations will succeed when performed. Due to themplus
semantics, the first one will always be selected.

the radio button in front of itis indeed selected (the user influences this during the
form’s lifetime).

The nice thing is that we can express this behaviour quite elegantly in terms of
RefLinkoperations. We show this by defining the somewhat simpler disjoint form
combinatoralt, which is a specialisation ofradioGrid: it joins exactly two forms
(denotedrefToForm1 andrefToForm2). Assume that we can create a two-button
radio group, returning a reference valuerefRadio :: Ref IO Int to its current se-
lection, which can take values{0,1}. Now we can define aRefLinkvalue:

rl1 = refLink ref0 refRadio
where ref0 = Ref

{ val = return0
, app= λf → do { 0 ← f 0; return() }
}

In other words: we linkrefRadioto a reference ‘to the unchangeable number 0’,
whoseval is always 0, and whoseapp function only succeeds when the result
of the function application is 0. This has the effect that theupdateoperation in
rl1 always selects the topmost radio button (and succeeds), while its propagate
operation only succeeds when this radio button is selected.4

We then liftrl1 into a form, and join it with the first subformform1 using�,
which utilizes the default (conjunctive)mappendoperator:

tell rl1 � refToForm1 ref

This form has the desired properties: with anupdate, the radio button is only
selected if the value inform1 can be updated, and with apropagate, the value in
form1 is only propagated if the radio button is selected. We define and userl2 in
a similar way, producing a second form. We finish thealt combinator by joining
both forms with�±:

refToForm1 ‘alt‘ refToForm2 = λref →
liftIO . . . �= (laybutton1, laybutton2, refRadio) →

let
rl1 = . . . — see above
rl2 = . . .

in
(tell rl1 � refToForm1 ref) �± λlayform1 →
(tell rl2 � refToForm2 ref) �± λlayform2 →
return0 $ grid 5 5 [[laybutton1, layform1], [laybutton2, layform2]]

What we have omitted in the second line goes into too much implementation
detail; it is an IO action which creates the radio buttons, and returns the layout
valueslaybutton1 andlaybutton2, as well asrefRadio.

4Note that thisRefLinkdoes not use any model state!

The radioGrid combinator is a straightforward generalization ofalt for lists.
ThecheckRowcombinator is also very much likealt, but does not show its second
argument form. However, itdoesuse itsRefLink.5

1.5 SAFETY

The flexibility provided by compositional functional references has a downside:
by omitting reference values, duplicating them, or using them in the wrong places,
forms with strange behaviour can be constructed. We give some examples:

1. declare2λ(a,b)→ entry′ [] a

This form never shows or changes the second element of its subject type.

2. λa → row′ 5 [entry′ [] a, entry′ [] a]

This form shows its value twice, and only propagates the new value in the
control on the right.

3. declareJust$ entry′ []

This form is only updated if its model contains aJust xvalue (actually, this is
the desired behaviour if the form is part of a disjoint form).If it does not, all
forms in the same alternative of the surrounding ‘disjoint clause’ are prevented
from being updated (normally, there should be none).

4. radioGrid [entry′ [], entry′ []]

This form will always put its value in the upper entry control. However, it will
propagate values from whichever entry control has its radiobutton selected.

To prevent the construction of these forms, the programmer can follow some rules
such as:

• Every declared reference should be used exactly once.

• Every data constructor of a form’s subject type should be declared exactly
once.

• References declared outside a disjoint form must not be usedinside it.

Of course, it would be better if these rules would be enforcedautomatically, e.g.
by the type system. Future research should formalize these rules.

1.6 RELATED WORK

As far as we know, the idea of explicitly using a radio button grid to combine
forms in a disjoint way is new. The fact that some radio buttons make other el-
ements (ir)relevant is recognized, but existing declarative (web) form languages

5Its RefLinkis joined with arefLink between the valueFalseand the checkbox value.

have to go out of their way to specify this. InXForms[12], it is accomplished
by providing an element’srelevant property with a Boolean expression that
includes anXPath pointer to the radio button choice. InWASH/CGI[9], the pro-
grammer builds adecision tree(see [10]) to express which data to use when a
certain radio button is selected.

A simple disjoint form combinator is already introduced in the thesis[1] from
which FunctionalForms originated. However, this combinator always joins ex-
actly two alternative forms. If the subject types of the top form and bottom
form aret1 and t2, resp., then the subject type of the composite form is always
Either t1 t2. In other words, logic and layout are not separated like theyare
presently.

Compositional referenceswere introduced by Kagawa[5] as a means to com-
pose mutable data structures such as arrays. In our previouswork[1, 2] we used
them in a more simple form and with a different goal: to conceptually separate a
form’s subject type and its layout.

Closely related to compositional references arelenses[3], which are also pairs
of accessor and modificator functions. While our approach uses a lot of ‘little’
references throughout the program, lenses are combined into a big lens whichis
the program; this program specifies a bidirectional transformation between model
and view.

1.7 CONCLUSIONS AND FUTURE WORK

In this paper, we have identified two patterns for composing forms that edit values
of a disjoint union type. The first pattern involves a list of radio buttons, and the
second involves a check box. To support these patterns in theFunctionalForms
library, we have introduced several new combinators.

These patterns illustrate a novel view, in which a form itself can be seen as
‘disjoint’. To demonstrate the fertility of this view, we have shown that these
disjoint forms exhibit a cleaner separation between logic and user interface. This
makes them more flexible.

However, this flexibility comes at a price: the constructionof forms with un-
wanted behaviour is possible. Methods for preventing this have yet to be re-
searched.

As a further enhancement toFunctionalForms, defining forms for values of a
customHaskell type is made easier, using aTemplate Haskell macro. This brings
the library closer to real-life use. The library version discussed in this paper will
shortly be available for download.6

We hope to further develop the approach of programming with reference val-
ues. We believe that it can be used to construct a far wider range of interfaces in
a declarative way.

6http://www.cs.ru.nl/˜sandr/FunctionalForms

